

Pole se seznamem (anglicky ComboBox), je kombinovaný ovládací prvek sestávající z textového pole (samostatně jako TextBox) a seznamu (ListBox). Někdy mu také podomácku říkám „roletka“.

Umožňuje zadání vstupu ručně či výběr ze seznamu přednastavených položek.

Abychom mohli daný prvek nasadit na listu, musíme si nejprve zobrazit na Pásu karet kartu Vývojář.



Excel 2010 – karta Vývojář

Ve skupině Ovládací prvky máme možnost vložit dva typy těchto prvků – formulářový (historicky starší, jednodušší) a prvek ActiveX (podobný, ale ne identický s tím, jaký znáte z formulářů VBA).

Pole se seznamem (ovládací prvek formuláře)

Práci s ním jsem se pokusil shrnout v obrázku. Pozn. Slovo „formuláře“ nemá nic společného s novodobými formuláři UserForm, se kterými pracujeme níže.



Pole se seznamem (ovládací prvek formuláře)

Výběr probíhá přes tlačítko Vložit a samotné vykreslení pak s pomocí myšky. Pokud chcete prvek zarovnat do mřížky, přidržte klávesu ALT. Pro editaci a přesun jej vždy vybírejte pravým tlačítkem myši. Vlastnosti najdete v kontextovém menu pod volbou Formát ovládacího prvku. Otevřený dialog nese stejný název. Možnosti jsou vcelku strohé a zřejmé. Okno po úpravám zavřeme a před použitím prvku už stačí jen zbavit jej zaměření (klepneme myškou jinam).

Provázaná buňka kdo ví proč vrací pořadí vybrané položky a k jejímu obsahu nám dopomůže až vyhledávací funkce (např. INDEX). Spojení funguje obousměrně, změna čísla v buňce má za následek výběr odpovídající položky v prvku (nula a menší číslo prvek vyčistí).

U prvku formuláře nezměníte barevnost ani písmo, neumí zobrazit více sloupců a v seznamu není možné vyhledávat ani se přesouvat mezi položkami kolečkem myši. Prvku lze přiřadit makro (viz také pravé tlačítko myši). Ve VBA spadá tento prvek do kolekce Shapes (název se na listu ukazuje v Poli názvů).

1	Sub PrvekFormularePristup()
2	
3	Dim shpPrvek As Shape
4	
5	Dim strPropojenaBunka As String
6	Dim strZdrojDat As String
7	
8	Set shpPrvek = ActiveSheet.Shapes("stahovací 1")
9	
10	'vlastnosti pod ControlFormat nebo DrawingObject
11	strPropojenaBunka = shpPrvek.ControlFormat.LinkedCell
12	strZdrojDat = shpPrvek.ControlFormat.ListFillRange
13	
14	End Sub

Pole se seznamem (ovládací prvek ActiveX)

I zde by měl napovědět hodně obrázek.



Pole se seznamem (ovládací prvek ActiveX)

Výběr a vykreslení se děje stejně jako v předchozím případě. Excel se přitom přepíná do Režimu návrhu a v něm také probíhají veškeré úpravy. Stačí klepnout v Pásu karet na Vlastnosti a pohnat si v otevřeném dialogu s názvem Properties (znáte možná už z editoru VBA). Nejpodstatnější jsou pro tuto chvíli vlastnosti ListFillRange (zdroj dat) a LinkedCell (propojená buňka), na zbytek se podíváme později. Na rozdíl od prvku formuláře vrací tento prvek primárně obsah vybrané položky.

Makro je zpravidla vázané na událost Change tohoto prvku. Stačí na ComboBox poklepat v Režimu návrhu a Excel vás přepne do editoru VBA. Prvek je umístěn v modulu daného listu. Procedura níže je ukázkou toho, jak makrem vracet pořadí vybrané položky.

1	Private Sub ComboBox1_Change()
2	
3	With ComboBox1
4	
5	'zápis pořadí položky do buňky
6	'nebyla vybrána položka .. ListIndex = -1
7	'první položka ... ListIndex = 0
8	Range("C20").Value = .ListIndex + 1
9	
10	End With
11	
12	End Sub



ComboBox na listu v editoru VBA

Pokud máme hotovo, vypneme Režim návrhu a je to.

Pozn. Ani jeden z prvků (formuláře, ActiveX) neumí vybírat více položek naráz, k tomu účelu slouží ListBox (o něm zase někdy jindy). V rámci položek se nemohou objevit obrázky (nastudujte prvek Microsoft ImageComboBox). Grafický vzhled ComboBoxu pod VBA neřídí XP styly (nedoporučuji měnit manifest pro Excel.exe!).

Jak naplnit ComboBox daty

Nadále budeme hovořit už jen o prvku ActiveX na formuláři UserForm. Zdrojem dat pro seznamy nemusí být nutně oblast buněk na listu. Položky lze přidávat jednotlivě (v cyklu) přes metodu AddItem nebo hromadně přiřazením pole do vlastnosti List. Nejprve základy. Přidat ComboBox na formulář (UserForm) předpokládám nepředstavuje problém.



ComboBox na formuláři UserForm

Okno Properties najdete v panelu nástrojů pod View / Properties Windows (klávesa F4).

Hurá na příklady. Formuláře si můžete vyzkoušet v návaznosti na List1.

```

1 Private Sub UserForm_Initialize()
2
3
4 Dim i As Integer
5
6 Dim strZnaky As String
7
8 'ComboBox1
9 *****
10
11 With ComboBox1
12     'jednotlivé přidání položek
13     .AddItem "raz"
14     .AddItem "dva"
15     .AddItem "tři"
16
17 End With
18
19 'ComboBox2
20 *****
21
22 With ComboBox2
23     'jednotlivé přidání položek
24     .AddItem "10,25"
25     .AddItem "10,50"
26     .AddItem "10,75"
27
28 End With
29
30
31 'ComboBox3
32 *****
33
34 'přifazení položek z oblasti buněk
35 'uvedení Value nutné
36 ComboBox3.List = Range("B2:B4").Value
37
38 'ComboBox4
39 *****
40
41 'přifazení položek z pojmenované oblasti buněk
42 'uvedení Value nutné
43 'ColumnCount = 2 ... dvousloupcový seznam
44 'ColumnHeads = True ... hlavička (vždy z oblasti listu)
45 ComboBox4.List = Range("rngTabulka").Value
46
47 'ComboBox5
48 *****
49
50 Dim arrPoleArray
51
52 'pole položek
53 arrPoleArray = Array("jablko", "hruška", "švestka")
54
55 'přifazení pole do prvku
56 ComboBox5.List = arrPoleArray
57
58 'ComboBox6
59 *****
60
61 Dim arrStatickePole(1 To 3)
62
63 arrStatickePole(1) = "ano"
64 arrStatickePole(2) = "ne"
65 arrStatickePole(3) = "nevím"
66
67 'přifazení pole do prvku
68 ComboBox6.List = arrStatickePole
69
70 'ComboBox7
71 *****
72
73 'řetězec položek
74 strOddelenePolozky = "Spejbl,Hurvínek,Mánička"
75
76 'převedení do pole položek
77 arrOddelenePolozky = Split(strOddelenePolozky, ",")
78
79 'přifazení pole do prvku
80 ComboBox7.List = arrOddelenePolozky
81
82 'ComboBox8
83 *****
84
85 For i = 1 To 5
86
87     'přidání položky do prvku
88     ComboBox8.AddItem "Pokoj " & i
89
90 Next i
91
92 'ComboBox9
93 *****
94
95 strZnaky = "abcdefghijklmnopqrstuvwxyz"
96
97 For i = 1 To Len(strZnaky)
98
99     'přidání položky do prvku
100     ComboBox9.AddItem Mid$(strZnaky, i, 1)
101
102 Next i
103
104 End Sub
105
106 Private Sub ComboBox2_Change()
107
108     'přenos číselné hodnoty do listu
109     Range("B9") = Cdbl(ComboBox2.Value)
110
111 End Sub

```



Základní techniky plnění ComboBoxu

Následuje ukázka, jak naplnit prvek položkami času.

```

1 Private Sub UserForm_Initialize()
2
3     Dim dtmCasOd As Date
4     Dim dtmCasDo As Date
5
6     Dim dtmCasKrok As Date
7
8     Dim intKrokHodiny As Integer
9     Dim intKrokMinuty As Integer
10    Dim intKrokSekundy As Integer
11
12    Dim sngCasKrok As Single
13    Dim sngCasRozdil As Single
14
15    Dim lngPocetJednotek As Long
16
17    'dílčí jednotky kroku
18    intKrokHodiny = 0
19    intKrokMinuty = 30
20    intKrokSekundy = 0
21
22    'čas
23    dtmCasKrok = TimeSerial(intKrokHodiny, intKrokMinuty, intKrokSekundy)
24    sngCasKrok = 24 * dtmCasKrok
25
26    'počet jednotek za den
27    lngPocetJednotek = 24 / sngCasKrok
28
29    'ComboBox1
30    '*****
31
32    'čas
33    For i = 1 To lngPocetJednotek
34        'přidání položky do prvku
35        ComboBox1.AddItem Format((i - 1) * dtmCasKrok, "Hh:Nn")
36    Next i
37
38    'ComboBox2
39    '*****
40
41    'doba trvání (přes 24 hodin)
42    For i = 1 To lngPocetJednotek
43        'přidání položky do prvku
44        ComboBox2.AddItem WorksheetFunction.Text(i * dtmCasKrok, "[hh]:mm")
45    Next i
46
47    'ComboBox3
48    '*****
49
50    'časový interval (např. směna) od-do
51    dtmCasOd = TimeSerial(6, 0, 0)
52    dtmCasDo = TimeSerial(14, 0, 0)
53
54    'dílčí jednotky kroku
55    intKrokHodiny = 0
56    intKrokMinuty = 15
57    intKrokSekundy = 0
58
59    'čas
60    dtmCasKrok = TimeSerial(intKrokHodiny, intKrokMinuty, intKrokSekundy)
61    sngCasKrok = 24 * dtmCasKrok
62    sngCasRozdil = 24 * (dtmCasDo - dtmCasOd)
63
64    'počet jednotek v intervalu
65    lngPocetJednotek = sngCasRozdil / sngCasKrok
66
67    'čas
68    For i = 1 To (lngPocetJednotek + 1)
69        'přidání položky do prvku
70        ComboBox3.AddItem Format(dtmCasOd + (i - 1) * dtmCasKrok, "Hh:Nn")
71    Next i
72
73 End Sub
74
75 Private Sub ComboBox1_Change()
76
77     'vložení hodnoty do buňky
78     Range("B13").Value = ComboBox1.Value
79
80 End Sub
81
82 Private Sub ComboBox2_Change()
83
84     'vložení hodnoty do buňky
85     Range("B13").Value = ComboBox2.Value
86
87 End Sub
88
89 Private Sub ComboBox3_Change()
90
91     'vložení hodnoty do buňky
92     Range("B13").Value = ComboBox3.Value
93
94 End Sub

```



Plnění ComboBoxu - časové údaje

Když jsem uvedl čas, nemůže chybět ani datum. Tentokrát jsem zvolil praktickou ukázkou s výběrem roku (ComboBox 1), měsíce (ConboBox2) a tomu odpovídajícímu počtu datumů (ComboBox3). Bonusově jsem poté přihodil k pondělním datumům ještě ISO číslo týdne roku a další vychytávky.

```

1 Private Sub UserForm_Initialize()
2
3 Dim intRok As Integer
4
5 intRok = Year(Now)
6
7 With cboRok
8
9 'naplnění prvku roky
10 'tento rok a minulý rok
11 .AddItem intRok - 1
12 .AddItem intRok
13
14 'předvýběr druhé položky
15 .ListIndex = 1
16
17 End With
18
19 With cboMesic
20
21 'naplnění prvku číselnými hodnotami
22 'tze řešit v cyklu
23 'zde s využitím metody Evaluate a vzorce listu
24 .List = [ROW(1:12)]
25
26 'předvýběr položky aktuálního měsíce
27 .ListIndex = Month(Now) - 1
28
29 End With
30
31 With cboMesicDatумы
32
33 'předvýběr položky aktuálního dne
34 .ListIndex = Day(Now) - 1
35
36 End With
37
38 End Sub
39
40 Private Sub cboMesic_Change()
41
42 'volání procedury Zmena
43 Call Zmena
44
45 End Sub
46
47 Private Sub cboRok_Change()
48
49 'volání procedury Zmena
50 Call Zmena
51
52 End Sub
53
54 Private Sub CommandButton1_Click()
55
56 'vlození hodnoty do buňky
57 Range("B11") = CDate(Left(cboMesicDatумы.Text, 10))
58
59 End Sub
60
61 Sub Zmena()
62
63 Dim i As Integer
64 Dim intRok As Integer
65 Dim intMesic As Integer
66 Dim intMesicPocetDni As Integer
67 Dim intTydenRoku As Integer
68
69 Dim dtmDate As Date
70
71
72 'v případě chyby (neúplná data při inicializaci formuláře)
73 'ukončit proceduru
74 On Error GoTo Konec
75
76 'rok
77 intRok = cboRok.Value
78
79 'měsíc
80 intMesic = cboMesic.Value
81
82 'počet dní (trik)
83 intMesicPocetDni = Day(DateSerial(intRok, intMesic + 1, 0))
84
85 With cboMesicDatумы
86
87 'vymazání stávajících položek
88 .Clear
89
90 For i = 1 To intMesicPocetDni
91
92 dtmDate = DateSerial(cboRok.Value, intMesic, i)
93
94 If Weekday(dtmDate, vbMonday) > 1 Then
95
96 'přidání datumu do prvku
97 .AddItem Format(dtmDate, "dd.mm.yyyy")
98
99 Else
100
101 'týden roku (ISO)
102 intTydenRoku = DatePart("ww", dtmDate, vbMonday, _
103 vbFirstFourDays)
104
105 'přidání datumu a týdne roku do prvku
106 .AddItem Format(dtmDate, "dd.mm.yyyy") & " (" & intTydenRoku & _
107 ")")
108
109 End If
110
111 Next i
112
113 'předvýběr první položky
114 .ListIndex = 0
115
116 End With
117
118 Konec:
119
120 End Sub

```




Plnění ComboBoxu - datum

Pokračujeme závislými seznamy otevřených sešitů a listů v nich.

```

1  Private Sub UserForm_Initialize()
2
3      Dim wkbSesit As Workbook
4
5      'ComboBox1
6      '*****
7
8      'pro každý otevřený sešit
9      For Each wkbSesit In Application.Workbooks
10
11         'přidání názvu sešitu do prvku
12         cboSesity.AddItem wkbSesit.Name
13
14         Next wkbSesit
15
16         'předvýběr aktivního sešitu
17         cboSesity.Value = ActiveWorkbook.Name
18
19     End Sub
20
21     Private Sub cboSesity_Change()
22
23         Dim shtList As Object
24
25         Dim strPolozka As String
26
27         'vybraná položka (sešit)
28         strPolozka = cboSesity.Value
29
30         'pro prvek s listy
31         With cboListy
32
33             'vymazání stávajících položek
34             .Clear
35
36             'pro každý list v aktivním sešitu
37             For Each shtList In Workbooks(strPolozka).Sheets
38
39                 'přidání názvu listu do prvku
40                 .AddItem shtList.Name
41
42                 Next shtList
43
44                 'je vybrán aktivní sešit?
45                 If strPolozka = ActiveWorkbook.Name Then
46
47                     'předvýběr aktivního listu
48                     .Value = ActiveSheet.Name
49
50                 Else
51
52                     'předvýběr první položky
53                     .ListIndex = 0
54
55                 End If
56
57             End With
58
59     End Sub

```



Plnění ComboBoxu - sešity a listy

Když jsme načali sešity, zůstaňme ještě chvíli u výpisu souborů ze složky vybrané uživatelem. Nebude chybět filtrování přípony, ale vynecháme rekurzivní procházení podsložek.

```

1 Private Sub Image1_Click()
2
3 Dim strSlozka As String
4 Dim arrPoleSouboru
5
6 'viceúcelový dialog, zde pro výběr složky
7 With Application.FileDialog(msoFileDialogFolderPicker)
8
9     'titulek, běžně "Procházet"
10    .Title = "Výběr složky"
11
12    'výchozí styl zobrazení, zde náhledy obsahu složek
13    'Windows 7, 64 bit, nefunkční
14    '.InitialView = msoFileDialogViewLargeIcons
15
16    'vícenásobný výběr složek nelze použít
17    '.AllowMultiSelect = True
18
19    'výchozí zobrazená složka, zde složka tohoto souboru
20    '.InitialFileName = ThisWorkbook.Path
21
22    'popis tlačítka, běžně "OK"
23    .ButtonName = "Vybrat"
24
25    'zobrazení dialogu
26    .Show
27
28    'byla vybrána složka?
29    If .SelectedItems.Count > 0 Then
30
31        'převzetí názvu složky
32        strSlozka = .SelectedItems(1)
33
34        End If
35
36    End With
37
38    'načtení názvů souborů s odpovídající příponou (maska pro sešity)
39    'ze složky do pole a jejich abecední seřazení
40    arrPoleSouboru = epfSlozkaSouborny(strSlozka & "*", "*.xl*")
41
42    'bylo vráceno pole?
43    If IsArray(arrPoleSouboru) Then
44
45        'vyčištění ovládacího prvku
46        cboSouborny.Clear
47
48        'přifazení pole do prvku
49        cboSouborny.List = arrPoleSouboru
50
51    Else
52
53        'zpráva
54        MsgBox "Nebyly nalezeny žádné soubory.", vbInformation + vbOKOnly
55
56    End If
57
58 End Sub
59
60 Function epfSlozkaSouborny(Slozka As String, Filtr As String)
61
62 Dim arrPole()
63 Dim strSouborNazev As String
64 Dim iPocet As Integer
65
66 strSouborNazev = Dir(Slozka & Filtr)
67
68 Do While strSouborNazev <> ""
69     If strSouborNazev <> "" Then
70         iPocet = iPocet + 1
71         ReDim Preserve arrPole(1 To iPocet)
72         arrPole(iPocet) = LCase(strSouborNazev)
73     End If
74     strSouborNazev = Dir()
75 Loop
76
77 If Not Len(Join(arrPole, "")) = 0 Then
78     QuickSort arrPole, LBound(arrPole), UBound(arrPole)
79     epfSlozkaSouborny = arrPole
80 Else
81     epfSlozkaSouborny = ""
82 End If
83
84 End Function
85
86 Sub QuickSort(vArray As Variant, inLow As Long, inHi As Long)
87 Dim pivot As Variant
88 Dim tmpSwap As Variant
89 Dim tmpLow As Long
90 Dim tmpHi As Long
91 tmpLow = inLow
92 tmpHi = inHi
93 pivot = vArray((inLow + inHi) \ 2)
94 While (tmpLow <= tmpHi)
95     While (vArray(tmpLow) < pivot And tmpLow < inHi)
96         tmpLow = tmpLow + 1
97     Wend
98     While (pivot < vArray(tmpHi) And tmpHi > inLow)
99         tmpHi = tmpHi - 1
100    Wend
101    If (tmpLow <= tmpHi) Then
102        tmpSwap = vArray(tmpLow)
103        vArray(tmpLow) = vArray(tmpHi)
104        vArray(tmpHi) = tmpSwap
105        tmpLow = tmpLow + 1
106        tmpHi = tmpHi - 1
107    End If
108    Wend
109    If (inLow < tmpHi) Then QuickSort vArray, inLow, tmpHi
110    If (tmpLow < inHi) Then QuickSort vArray, tmpLow, inHi
111 End Sub

```



Plnění ComboBoxu – soubory ze složky

Jako zdroj pro ComboBox samozřejmě slouží i databáze. Zde krátká ukázka propojení z Microsoft Access.

```
1 Private Sub UserForm_Initialize()  
2  
3 Dim cn As Object  
4 Dim rs As Object  
5 Dim strSql As String  
6 Dim strConnection As String  
7  
8 'připojení do databáze (Microsoft Access)  
9  
10 'vytvoření objektu připojení  
11 Set cn = CreateObject("ADODB.Connection")  
12  
13 'řetězec pro připojení  
14 strConnection = "Provider=Microsoft.Jet.OLEDB.4.0;" & "Data Source=" & _  
15 ThisWorkbook.Path & "\databaze.mdb"  
16  
17 'SQL dotaz (obec začínající na Vče... a jeho PSČ)  
18 strSql = "SELECT DISTINCT Obec, PSC FROM PSC WHERE Obec Like 'Vče%'"  
19  
20 'připojení  
21 cn.Open strConnection  
22  
23 'recordset  
24 Set rs = cn.Execute(strSql)  
25  
26 'konverze (transpozice) záznamů do pole  
27 arrPoleObce = WorksheetFunction.Transpose(rs.GetRows)  
28  
29 'přiřazení pole do prvku  
30 ComboBox1.List = arrPoleObce  
31  
32 'uzavření spojení  
33 rs.Close  
34 cn.Close  
35  
36 'uvolnění paměti  
37 Set rs = Nothing  
38 Set cn = Nothing  
39  
40 End Sub
```



Plnění ComboBoxu – databáze

V případě databází se využívají prakticky všechny způsoby načítání položek, od AddItem v cyklu po přiřazení pole (transformujeme výstupy GetRows, případně GetString, zbavujeme záznamy položek Null, upravujeme datový typ apod.). Pokud jste nedočeti před zkoušením do konce, pak vězte, že vícesloupcové ComboBoxy doprovází vlastnost ColumnCount (zde ColumnCount = 2).

Zbývají ještě drobné tipy na vytváření seznamů bez nutnosti textových polí (názvy dnů, měsíců).

```
1 Private Sub UserForm_Initialize()  
2  
3     'ComboBox1  
4     '*****  
5  
6     'dny týdne slovně  
7     For i = 1 To 7  
8         'přidání položky do prvku  
9         '24.4.2016 ... neděle  
10        ComboBox1.AddItem Format(DateSerial(2016, 4, 24) + i, "dddd")  
11    Next i  
12  
13    'ComboBox2  
14    '*****  
15  
16    'měsíce slovně  
17    For i = 1 To 12  
18        'přidání položky do prvku  
19        ComboBox2.AddItem Format(DateSerial(2016, i, 1) + i, "mmmm")  
20    Next i  
21  
22    'ComboBox3  
23    '*****  
24  
25    'měsíce slovně (německy)  
26    For i = 1 To 12  
27        'přidání položky do prvku  
28        ComboBox3.AddItem WorksheetFunction.Text(DateSerial(2016, i, 1) + i, _  
29            "[$-407]mmmm")  
30    Next i  
31  
32 End Sub
```



Plnění ComboBoxu – dny týdne a názvy měsíců
jinak...

Do sytosti jsme si pohráli a ještě zbývá vážnější zastavení se u vybraných vlastností ComboBoxu.

Vybrané vlastnosti (vícesloupcového) ComboBoxu



ComboBox – více sloupců a vybrané vlastnosti

BoundColumn ... číslo sloupce, ze kterého se bude brát hodnota Value (1 .. první sloupec, i skrytý)

ColumnsCount ... určuje, kolik sloupců dat bude ComboBox obsahovat (včetně skrytých), pro skrytý sloupec je ColumnWidths = 0)

ColumnWidths ... šířky sloupců v bodech (palcích, centimetrech) oddělených středníky (např. 0 pt;111 pt;109 pt), pro skrytý sloupec se zapisuje 0, šířku posledního sloupce lze vynechat (součet šířek nesedí přesně s vlastností Width prvku, teoreticky $Width = \text{součet šířek} + ColumnsCount - 1$)

LinkedCell ... adresa propojené buňky (např. List1!D20), platí pro prvek ComboBox na listu

ListRows ... počet viditelných položek po rozbalení seznamu bez nutnosti užití posuníku

ListWidth ... šířka seznamu po rozbalení

MatchEntry ... ponechávejte na 1-fmMatchEntryComplete, vyhledávání existující položky bude tak probíhat v souladu s posloupností zadávaných znaků v textovém poli, resp. při aktivaci prvku.

RowSource ... adresa zdroje dat braného z listu (např. List1!G3:I6), platí pro prvek na formuláři UserForm, pro prvek ComboBox na listu jde o vlastnost ListFillRange

Style ... pokud uživatel nemá mít možnost zapsat do textového pole prvku vlastní hodnotu a bude pouze vybírat z existujících položek, zvolte 2-fmStyleDropDownList

TextColumn ... ,číslo sloupce, jehož položky se objevují po výběru ze seznamu v textovém poli a ve vlastnosti Text,

TextColumn = -1 ... zobrazí hodnotu 1. sloupce s hodnotou ColumnWidths > 0, tj. prvního viditelného

TextColumn = 0 ... zobrazí hodnotu vlastnosti ListIndex

TextColumn = 1 ... zobrazí hodnotu prvního sloupce (i skrytého)

Na závěr

Jsem si vědom toho, že jsem po celou dobu článku ignoroval vestavěný prvek na listu Excelu, který je jakýmsi hybridem mezi ComboBoxem a ListBoxem – seznam dostupný pod kartou Data / Ověření. V podstatě představuje jednovýběrový ListBox a zaslouží si samostatnou kapitolu a článek... Padlááá.

Příloha

[excel_ovladaci_prvky_combobox.zip](#)